

# Programando juegos en forth/r3

r3 es un lenguaje de programación derivado del ColorForth.

<https://github.com/phreda4/r3>

Pablo H. Reda

2023

Este manual esta hecho con la colaboración de alumnos:

Vladimir Gomez Perez

Bianca Cipollone Di Croce

# Programando un Juego en r3

Para hacer programas con gráficos necesitamos acceso a esta capacidad, para esto, vamos a utilizar la biblioteca SDL: <https://www.libsdl.org/>

Las palabras que manejan esta biblioteca las encontramos en SDL2GFX.r3

Este listado es la base de cualquier programa gráfico.

1	^r3/win/sdl2gfx.r3	incluye las palabras gráficas
2		
3	:juego	
4	0 SDLcls	limpia la ventan con color 0 (negro)
5	SDLredraw	refresca la ventana
6	SDLkey	apila la tecla
7	>esc<=? ( exit )	si soltó escape... sale
8	drop ;	
9		
10	:main	
11	"r3sdl" 640 480 SDLinit	inicia ventana de 640 x 480
12	'juego SDLshow	repite palabra juego hasta exit
13	SDLquit ;	quita la ventana
14		
15	: main ;	

Este código va a crear una ventana.

El programa empieza a ejecutarse en la línea 15, donde se ejecuta la palabra main.

En la línea 11 hace una ventana gráfica de 640 por 480 píxeles.

La línea 12 usa **SDLshow** para hacer un loop con esta palabra, se necesita la dirección de la palabra y no la palabra, hasta que, dentro de esta palabra, se llama a **exit**.

La línea 4 limpia la ventana con el color 0, negro, entre esta palabra y la siguiente podemos dibujar los graficos que queramos hasta la línea 5 donde redibuja la ventana.

La línea 6 obtengo la tecla del usuario.

en la línea 7 comparo si se soltó la tecla ESC y ejecutó **exit** si ocurre esto.

## Dibujos del juego

Con esta base empezaremos por crear una hoja de sprites, esto es, un archivo con todos los gráficos que usaremos en el juego. Utilizaremos sprites de 16 x 16 píxeles.



Cada dibujo va a numerarse de arriba a abajo y de izquierda a derecha, comenzando de 0, tenemos 9 sprites.

Comenzamos definiendo una variable para guardar los sprites, y las coordenadas para dibujar  
En la línea 19 utilizamos para cargar el archivo gráfico:

```
::ssload | w h "file" -- sprite
```

El ancho (w) y el alto (h) de cada sprite en la hoja y la ruta completa con nombre del archivo (file) son los parámetros que necesita SSLOAD, dejando la dirección del gráfico cargado en memoria (un 0 significa que no pudo cargar el archivo). Este valor lo guardamos en una variable y vamos a utilizarlo cuando dibujemos en pantalla.

Disponemos de varias palabras para dibujar en pantalla el sprite:

```
::ssprite | x y n ssprite --
```

para dibujar el sprite N (comienza de 0) en la posición X Y

```
::sspriter | x y ang n ssprite --
```

para dibujar el sprite rotado en el ángulo ANG: 0.5 es media vuelta, 0.25 son 90 grados, etc.

```
::sspritez | x y zoom n ssprite --
```

para dibujar el sprite con ZOOM, 2.0 será el doble de grande, 0.5 será la mitad, etc.

```
::sspriterz | x y ang zoom n ssprite --
```


para dibujar el sprite con ZOOM y ANG a la vez

En la línea 8, elegimos dibujar el sprite con zoom, X e Y son la posición de píxeles que será el centro de nuestro sprite, el nivel de zoom, aquí un número con punto decimal llamado número de punto fijo..donde podremos achicar o agrandar el dibujo, en nuestro caso lo duplicaremos, luego el número de sprite y por último la hoja donde se encuentra, que es donde lo cargamos.

El código se encuentra en la distribución del lenguaje:

1	^r3/win/sdl2gfx.r3
2	
3	#sprites
4	#x 320 #y 380
5	
6	:juego
7	0 SDLcls
8	x y 2.0 0 sprites sspritez
9	SDLredraw
10	
11	SDLkey
12	>esc< =? ( exit )
13	<le> =? ( -1 'x +! )
14	<ri> =? ( 1 'x +! )

15	drop ;
16	
17	:main
18	"r3sdl" 640 480 SDLinit
19	16 16 "media/img/manual.png" ssload 'sprites !
20	'juego SDLshow
21	SDLquit ;
22	
23	: main ;



## Mejorando el movimiento

En el código, las flechas izquierda y derecha restan y suman a la variable X, donde guardamos la posición de la nave.

Si probamos el programa notamos que el efecto de movimiento no es el deseado, necesitamos que mientras esté presionada la tecla, la nave se mueva y para esto necesitamos tener variables de velocidad y sumar constantemente esta velocidad a la posición.

Para que las velocidad tenga la posibilidad de ser una parte del entero vamos a utilizar números de punto fijo, para esto cuando lo indicamos debemos agregar obligatoriamente el punto decimal, y necesitamos convertirlo a entero cuando indicamos el pixel en pantalla por lo que utilizaremos la palabra

```
::int. | punto.fijo -- entero
```

Utilizamos **<le>** (tecla LEft o flecha a la izquierda) para detectar cuando se oprime la tecla y **>le<** cuando se suelta la tecla, ya que necesitamos cambiar la velocidad en estos dos estados.

Creamos una nueva palabra para separar el código que maneja el jugador, y acá dibujamos el sprite y sumamos las posiciones con las velocidades.

Probando el código notaremos la diferencia en el funcionamiento.

1	^r3/win/sdl2gfx.r3
2	
3	#sprites
4	#x 320.0 #y 380.0

```

5  #xv 0 #yv 0
6
7  :jugador
8      x int. y int. 2.0 0 sprites sspritez
9      xv 'x +! yv 'y +! ;
10
11 :juego
12     0 SDLcls
13     jugador
14     SDLredraw
15
16     SDLkey
17     >esc< =? ( exit )
18     <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
19     <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
20     drop ;
21
22 :main
23     "r3sdl" 640 480 SDLinit
24     16 16 "media/img/manual.png" ssload 'sprites !
25     'juego SDLshow
26     SDLquit ;
27
28 : main ;

```

## Llegaron los extraterrestres

Agregamos otro elemento al programa, el alien, necesitamos la posición en **XA** y **YA**, y moverlo, no con las teclas, sino que sumando constantemente a **YA**, que va a ser que baje en la pantalla.

La línea 11 reinicia el alien, **XA** será un número al azar entre 0 y el ancho de pantalla, en punto fijo.

La línea 16 va a reiniciar el alien cuando salga por abajo, o sea, la coordenada **YA** sea mayor al alto de la ventana.

La línea 24 primero dibuja el jugador y luego el alien.

```

^r3/win/sdl2gfx.r3
^r3/lib/rand.r3

#sprites
#x 320.0 #y 380.0
#xv 0 #yv 0

#xa 0.0 #ya 100.0

:+alien
    -16.0 'ya ! 640.0 randmax 'xa ! ;

:alien
    xa int. ya int. 2.0 2 sprites sspritez
    2.0 'ya +!
    ya 480.0 >? ( +alien ) drop ;

```

```

:jugador
    x int. y int. 2.0 0 sprites sspritez
    xv 'x +! yv 'y +! ;

:juego
    0 SDLcls
    jugador alien
    SDLredraw

    SDLkey
    >esc< =? ( exit )
    <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
    <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
    drop ;

:main
    "r3sdl" 640 480 SDLinit
    16 16 "media/img/manual.png" ssload 'sprites !
    'juego SDLshow
    SDLquit ;

: main ;

```

## Disparos

Agregamos otras dos posiciones para guardar las coordenadas del disparo. Necesitamos indicar cuando está presente el disparo y para esto, en este caso, elegimos utilizar la variable XD, cuando sea negativa será que no hay disparo. Solamente podemos tener un disparo en pantalla.

La palabra disparo va a dibujar la bala y modificar su posición. la línea 26 consulta sobre la variable xd y si esta es negativa, termina la palabra, esto es, no hace mas nada aqui.

La línea 27 se encarga de dibujar el disparo.

La línea 28 resta 4 a la posición en Y, quiere decir que irá subiendo en pantalla.

La línea 29 pregunta si la posición en YD es negativo, o sea, sale de pantalla, si ocurre esto pone en negativo el X para que no entre a esta palabra más, por la línea 26.

La palabra +disparo primero controla que no haya una bala en la pantalla, y si es posible, copiar las coordenadas de nuestra nave en las coordenadas de la bala para que funcione este mecanismo.

La línea 44 pregunta si se presiona la tecla ESCAPE y ejecuta +disparo cuando esto ocurre.

```

^r3/win/sdl2gfx.r3
^r3/lib/rand.r3

#sprites
#x 320.0 #y 380.0
#xv 0 #yv 0

#xa 0.0 #ya 100.0

:+alien

```

```

-16.0 'ya ! 640.0 randmax 'xa ! ;

:alien
  xa int. ya int. 2.0 2 sprites sspritez
  2.0 'ya +!
  ya 480.0 >? ( +alien ) drop ;

:jugador
  x int. y int. 2.0 0 sprites sspritez
  xv 'x +! yv 'y +! ;

#xd #yd

:disparo
  $ffffff SDLColor
  xd -? ( drop ; ) | si es negativo sale
  int. 1 - yd int. 3 8 SDLFRect
  -4.0 'yd +!
  yd -? ( -1.0 'xd ! ) drop ;

:+disparo
  xd +? ( drop ; ) drop
  x 'xd ! y 'yd ! ;

:juego
  0 SDLcls
  disparo jugador alien
  SDLredraw

  SDLkey
  >esc< =? ( exit )
  <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
  <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
  <esp> =? ( +disparo )
  drop ;

:main
  "r3sdl" 640 480 SDLinit
  16 16 "media/img/manual.png" ssload 'sprites !
  'juego SDLshow
  SDLquit ;

: main ;

```

## Colisión y contar puntos

Definimos una variable para almacenar los puntos en la línea 35. Vamos a agregar la detección de la colisión entre la bala y el alien.

Para esto, en la línea 41, vamos a calcular la distancia que hay en x entre estas dos variables XA y XD, restando y obteniendo el valor absoluto y calculando que si hay más 16 pixeles, no va a producirse el choque.

Vamos a hacer lo mismo para la coordenada Y, en la línea 42, pero aquí calcularemos que la distancia sea de 8 pixeles, esto es, porque el dibujo del alien es mas ancho que alto, variando estos numeros sera mas dificil o mas facil pegarle.

Esto se realiza en la palabra CHOCO?, donde si se cumplen las condiciones, entonces sumaremos 1 al puntaje, borraremos el disparo y agregaremos un nuevo alien, borrando el anterior.

La línea 48 agrega al bucle principal esta comprobación del choque de la bala y el alien.

```
^r3/win/sdl2gfx.r3
^r3/lib/rand.r3

#sprites
#x 320.0 #y 380.0
#xv 0 #yv 0

#xa 0.0 #ya 100.0

:+alien
    -16.0 'ya ! 640.0 randmax 'xa ! ;

:alien
    xa int. ya int. 2.0 2 sprites sspritez
    2.0 'ya +!
    ya 480.0 >? ( +alien ) drop ;

:jugador
    x int. y int. 2.0 0 sprites sspritez
    xv 'x +! yv 'y +! ;

#xd #yd

:disparo
    $ffffff SDLColor
    xd -? ( drop ; ) | si es negativo sale
    int. 1 - yd int. 3 8 SDLFRect
    -4.0 'yd +!
    yd -? ( -1.0 'xd ! ) drop ;

:+disparo
    xd +? ( drop ; ) drop
    x 'xd ! y 'yd ! ;

#puntos 0

:+punto
    1 'puntos +! -1 'xd ! +alien ;

:choco?
    xa xd - abs 16.0 >? ( drop ; ) drop
    ya yd - abs 8.0 >? ( drop ; ) drop
    +punto ;

:juego
    0 SDLcls
    disparo jugador alien
```



	<pre> choco? SDLredraw  SDLkey &gt;esc&lt;=? ( exit ) &lt;le&gt;=? ( -1.0 'xv ! ) &gt;le&lt;=? ( 0 'xv ! ) &lt;ri&gt;=? ( 1.0 'xv ! ) &gt;ri&lt;=? ( 0 'xv ! ) &lt;esp&gt;=? ( +disparo ) drop ;  :main   "r3sdl" 640 480 SDLinit   16 16 "media/img/manual.png" ssload 'sprites !   'juego SDLshow   SDLquit ;  : main ; </pre>
--	--

## Perder también hay que programarlo

Para que sea totalmente funcional falta agregar una variable de vidas, en la línea 37 y agregar cuando el alien choca a nuestra nave.

La forma de hacerlo es igual. Definiendo la palabra perdio? en la línea 51 y 52 calculamos que tanto en la coordenadas X como en Y la distancia entre las coordenadas del jugador y el alien sean menores a un umbral definido. Cuando esto ocurre se resta la cantidad de vidas y se comprueba que las vidas sean 0 para salir del programa llamado a la palabra exit.

las líneas 55 y 56 retornan la nave al inicio y agrega un nuevo alien, quitando el anterior para que continúe el juego.

	<pre> ^r3/win/sdl2gfx.r3 ^r3/lib/rand.r3  #sprites #x 320.0 #y 380.0 #xv 0 #yv 0  #xa 0.0 #ya 100.0  :+alien   -16.0 'ya ! 640.0 randmax 'xa ! ;  :alien   xa int. ya int. 2.0 2 sprites sspritez   2.0 'ya +!   ya 480.0 &gt;? ( +alien ) drop ;  :jugador   x int. y int. 2.0 0 sprites sspritez   xv 'x +! yv 'y +! ;  #xd #yd </pre>
--	--

```

:disparo
    $ffffff SDLColor
    xd -? ( drop ; ) | si es negativo sale
    int. 1 - yd int. 3 8 SDLFRect
    -4.0 'yd +!
    yd -? ( -1.0 'xd ! ) drop ;

:+disparo
    xd +? ( drop ; ) drop
    x 'xd ! y 8.0 -
    'yd ! ;

#puntos 0
#vidas 3

:+punto
    1 'puntos +!
    -1 'xd !
    +alien ;

:choco?
    xa xd - abs 16.0 >? ( drop ; ) drop
    ya yd - abs 8.0 >? ( drop ; ) drop
    +punto
    ;

:perdio?
    xa x - abs 16.0 >? ( drop ; ) drop
    ya y - abs 16.0 >? ( drop ; ) drop
    -1 'vidas +!
    vidas 0? ( exit ) drop
    320.0 'x !
    380.0 'y !
    +alien
    ;

:juego
    0 SDLcls
    disparo jugador    alien
    choco? perdio?
    SDLredraw

    SDLkey
    >esc< =? ( exit )
    <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
    <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
    <esp> =? ( +disparo )
    drop ;

:main
    "r3sdl" 640 480 SDLinit
    16 16 "media/img/manual.png" ssload 'sprites !
    +alien
    'juego SDLshow
    SDLquit ;

```

	: main ;
--	----------

## Mostrar texto

Para utilizar texto en una ventana gráfica, tenemos varias posibilidades. Veamos como unas de las más simples. Incluyendo a la biblioteca bfont, tenes dos fuentes de ancho fijo, en la línea 79 elegimos la fuente.

Usamos para ubicar dónde se va a imprimir el texto

bat | x y --

para imprimir un texto

bprint | "" --

tener en cuenta que bprint no produce el texto con los modificadores de %, por eso acá usamos sprint para convertir a un texto con valores de la pila.

Las líneas 66 y 67 preparan el texto para mostrar vidas y puntos y lo imprimen en el lugar indicado por bat

..	
3	^r3/util/bfont.r3
..	
66	10 8 bat puntos "puntos:%d" sprint bprint
67	10 24 bat vidas "vidas:%d" sprint bprint
..	
79	bfont1
..	



## Estrellas de fondo

Vamos a agregar un fondo de estrellas que se mueven al juego utilizando un buffer de memoria. Necesitamos guardar posiciones X e Y para cada estrella y vamos a dibujar un pixel en estos lugares.

Para guardar estas posiciones utilizamos una parte de memoria, y elegimos guardar los números en 16 bits, 2 bytes, por eso utilizamos w!+ y w@+.

La palabra +estrella agrega las coordenadas X e Y en memoria en orden inverso dibuja.estrellas recorre en buffer y extrae estas coordenadas y llama a dibujar cada punto

Para mover las estrellas, en este caso vamos a incrementar en 1 la coordenada Y, cuando supera el límite de la pantalla (variable SH) vuelve a 0 para que aparezca por el otro lado de la ventana.

Por último tenemos una palabra para agregar las estrellas que necesitamos, SW es el ancho de la ventana, SH es el alto, **RANDMAX** produce un número aleatorio entre 0 y el valor de la pila.

Como indicamos un tamaño de 1024 bytes y cada estrella ocupa 4, 2 para x y 2 para Y, podremos guardar 256 coordenadas, una parte de la programar consiste en definir límites y cantidades Necesitamos agregar el dibujo y movimiento de las estrellas al bucle principal, en la línea 84 y rellenar las coordenadas al inicio del programa

```
^r3/win/sdl2gfx.r3
^r3/lib/rand.r3
^r3/util/bfont.r3

#buffer * 1024 | lugar para guardar las coordenadas
#buffer> 'buffer

:+estrella | x y --
    buffer> w!+ w!+ 'buffer> ! ;

:.estrellas
$ffffff sdlColor
    'buffer ( buffer> <?
        w@+ swap
        over 1 + sh >? ( 0 nip )
        over 2 - w!
        w@+ rot
        SDLPoint ) drop ;

:llenaestrellas
    256 ( 1? 1 -
        sw randmax sh randmax +estrella
    ) drop ;

#sprites
#x 320.0 #y 380.0
#xv 0 #yv 0

#xa 0.0 #ya 100.0

:+alien
    -16.0 'ya ! 640.0 randmax 'xa ! ;

:alien
```

```

    xa int. ya int. 2.0
    msec 7 >> $3 and 2 +
    sprites sspritez
    2.0 'ya +!
    ya 480.0 >? ( +alien ) drop ;

:jugador
    x int. y int. 2.0
    msec 7 >> $1 and
    sprites sspritez
    xv 'x +! yv 'y +! ;

#xd #yd

:disparo
    $ffffff SDLColor
    xd -? ( drop ; ) | si es negativo sale
    int. 1 - yd int. 3 8 SDLFRect
    -4.0 'yd +!
    yd -? ( -1.0 'xd ! ) drop ;

:+disparo
    xd +? ( drop ; ) drop
    x 'xd ! y 'yd ! ;

#puntos 0
#vidas 3

:+punto
    1 'puntos +!
    -1 'xd !
    +alien ;

:choco?
    xa xd - abs 16.0 >? ( drop ; ) drop
    ya yd - abs 8.0 >? ( drop ; ) drop
    +punto ;

:perdio?
    xa x - abs 16.0 >? ( drop ; ) drop
    ya y - abs 16.0 >? ( drop ; ) drop
    -1 'vidas +!
    vidas 0? ( exit ) drop
    320.0 'x !
    380.0 'y !
    +alien ;

:juego
    0 SDLcls
    .estrellas
    disparo jugador    alien
    choco? perdio?
    10 8 bat puntos "puntos:%d" sprint bprint
    10 24 bat vidas "vidas:%d" sprint bprint
    SDLredraw

```

```

SDLkey
>esc< =? ( exit )
<le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
<ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
<esp> =? ( +disparo )
drop ;

:main
  "r3sdl" 640 480 SDLinit
  bfont1
  16 16 "media/img/manual.png" ssload 'sprites !
  llenaestrellas
  +alien
  'juego SDLshow
  SDLquit ;

: main ;

```



## Animar

Los sprites de la hoja consisten en una imagen con las diferentes dibujos que representan animaciones, generalmente se trata de secuencias que representan caminar, correr, saltar, esperar para personajes o estado para objetos como, vaso lleno o vaso vacío.

La forma de utilizar estos gráficos es, primero, organizar estas animaciones en una grilla regular, es decir, todos los casilleros del mismo tamaño. Este tamaño se indica cuando se carga la hoja. Tenemos palabras que van a dibujar este sprite, indicando el número de gráfico, este número comienza en 0 y va de izquierda a derecha y de arriba a abajo.

Para animar lo que necesitamos es alternar este número en el tiempo, utilizamos una palabras construidas para este fin

## Muchos sprites

A veces necesitamos dibujar muchas instancias de un determinado sprite.

Para esto usaremos una biblioteca que permite manejar una porción de memoria como lista de multi celdas con una capacidad de 16 números (celdas).

```
^r3/util/arr16.r3
```

Este módulo está construido para realizar exactamente lo que necesitamos.

Primero, reservar la cantidad de memoria necesaria para contener un máximo indicado por el programador, y necesitamos una variable para indicar a qué lista hacemos referencia.

Se necesita una variable con dos valores para guardar este mecanismo

```
#listalien 0 0
```

para inicializar vamos a definir cuál es la capacidad máxima de esta lista

```
20 'listalien p.ini
```

Esta palabra genera en memoria una estructura con la siguiente forma:

	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
0																
1																
2																
...																
19																

Las filas son los elementos y las columnas las celdas que tiene cada elemento.

Cada elemento van a ser 16 celdas de memoria, en la primera vamos a poner una dirección de palabra especial que va a dibujar el sprite en pantalla y calcular la próxima posición.

Cada vez que se llame a esta palabra, va a recibir de la pila la dirección de memoria donde está el inicio de las 16 celdas y debe quitarla de la pila al finalizar la palabra.

Si queremos que este sprite desaparezca, vamos a dejar en la pila un 0 al salir de la ejecución por elemento.

```
:+alien | --
    'alien 'listalien p!+ >a | vector a palabra de ejecucion
    10.0 a!+ | x
    20.0 a!+ | y
    2.0 a!+ | zoom
    12 a!+ | nro sprite
    sprites a!+ | hoja de sprite
    ;
```

	0	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15
0	'alien	10.0	20.0	2.0	12	sprites										
1																
2																
...																
19																

```
:alien | adr --
    >a
    a@+ int. a@+ int. | x y , convertidos a enteros
    a@+                | zoom
    a@+                | nro de sprite
    a@+                | hoja de sprite
    sspritez ;
```

En estas celdas podemos guardar las variables que queramos, como la posición, la velocidad de desplazamiento, la animación del sprite, etc.

Estos valores se acceden o modifican por la posición relativa a la dirección que tenemos en la pila, Aunque si queremos podemos definir una palabra que indique a qué nos estamos refiriendo.

Es conveniente definir las posiciones consecutivas de las celdas como datos que vamos a necesitar para, por ejemplo, dibujar nuestro sprite, así, como estamos utilizando:

```
sspritez | x y z nro sprite -
```

Podemos usar ese orden para guardar los valores, usando el registro A podemos recorrer y usar estos valores, líneas 26 a 32.

Para dibujar constantemente en pantalla debemos agregar en el bucle de dibujo, antes de SDLredraw, línea 69.

Podríamos agregar estos elementos con f1 para probar, entonces cuando testeamos las teclas presionadas agregamos, línea 80.

Esto debería agregar dibujos a la pantalla

```
^r3/win/sdl2gfx.r3
^r3/lib/rand.r3
^r3/util/bfont.r3
^r3/util/arr16.r3

#sprites
#puntos 0
#vidas 3

#x 320.0 #y 380.0
#xv 0 #yv 0
```



```
#listalien 0 0 | lista de aliens
```

```
|----- Jugador
```

```
:jugador
```

```
    x int. y int. 2.0  
    msec 7 >> $1 and  
    sprites sspritez  
    xv 'x +! yv 'y +! ;  
    ;
```

```
|----- Aliens
```

```
:alien | adr -
```

```
    >a  
    a@+ int. a@+ int. | x y , convertidos a enteros  
    a@+ | zoom  
    a@+ | nro de sprite  
    a@+ | hoja de sprite  
    sspritez ;
```

```
:+alien
```

```
    'alien 'listalien p!+ >a  
    640.0 randmax a!+ | x  
    480.0 randmax a!+ | y  
    2.0 randmax 1.0 + a!+  
    12 randmax a!+  
    sprites a!+  
    ;
```

```
|----- fondo de estrellas
```

```
#buffer * 1024 | lugar para guardar las coordenadas
```

```
#buffer> 'buffer
```

```
:+estrella | x y -
```

```
    buffer> w!+ w!+ 'buffer> ! ;
```

```
:.estrellas
```

```
    $ffffff sdlColor  
    'buffer ( buffer> <?  
        w@+ swap  
        over 1 + sh >? ( 0 nip )  
        over 2 - w!  
        w@+ rot  
        SDLPoint ) drop ;
```

```
:llenaestrellas
```

```
    256 ( 1? 1 -  
        sw randmax sh randmax +estrella  
    ) drop ;
```

```
|----- Juego
```

```
:juego
```

```
    0 SDLcls  
    timer.  
    .estrellas  
    'listalien p.draw
```

```

jugador


10 8 bat puntos "puntos:%d" sprint bprint
10 24 bat vidas "vidas:%d" sprint bprint
SDLredraw

SDLkey
>esc< =? ( exit )
<le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
<ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
<f1> =? ( +alien )
drop ;

:main
  "r3sdl" 640 480 SDLinit
  bfont1
  16 16 "media/img/manual.png" ssload 'sprites !
  llenaestrellas
  200 'listalien p.ini
  timer<
  'juego SDLshow
  SDLquit ;

: main ;

```



Utilizando este recurso vamos a utilizar una lista de sprites para los disparos y otra para los aliens, aunque es posible utilizar una lista para ambos, a la vez, es conveniente separarlos porque cuando comprobemos si una bala chocó con un alien, esta separación nos simplifica el código ya para cada bala vamos a recorrer la lista de aliens. Si no está separado debería comprobar en cada recorrido el tipo de cada sprite.

```
^r3/win/sdl2gfx.r3
```

```

^r3/lib/rand.r3
^r3/util/bfont.r3
^r3/util/arr16.r3

#sprites

#puntos 0
#vidas 3

#x 320.0 #y 380.0
#xv 0 #yv 0

#listalien 0 0 | lista de aliens
#listshoot 0 0 | lista de disparos

:objsprite | adr -- adr
  dup >a
  a@+ int. a@+ int. | x y
  a@+ | zoom
  a@ timer+ dup a!+ | anima
  nanim a@+ sspritez
  dup 40 + @ over +! | vx
  dup 48 + @ over 8 + +! | vy
  ;

|----- Jugador
:jugador
  x int. y int. 2.0
  msec 7 >> $1 and
  sprites sspritez
  xv 'x +! yv 'y +!
  ;

:perdevida
  -1 'vidas +!
  vidas 0? ( exit ) drop
  320.0 'x !
  380.0 'y !
  ;

|----- Disparo
#hit
:choque | x y i n p -- x y p
  dup 8 + >a
  pick4 a@+ - pick4 a@+ -
  distfast 20.0 >? ( drop ; ) drop
  dup 'listalien p.del
  1 'puntos +!
  0 'hit !
  ;

:bala | v --
  objsprite

  1 'hit !

```

```

dup @ |x
over 8 + @ -20.0 <? ( 3drop 0 ; ) | y fuera de pantalla
'choque 'listalien p.mapv | 'vector list --
2drop
hit 0? ( nip ; ) drop
drop
;

:+disparo
'bala 'listshoot p!+ >a
x a!+ y 16.0 - a!+ | x y
1.0 a!+ | zoom
8 2 10 vci>anim | vel cnt ini
a!+ sprites a!+ | anim sheet
0 a!+ -3.0 a!+ | vx vy
;

|----- Alien
:alien | v --
  objsprite
  dup @ x -
  over 8 + @
  500.0 >? ( 3drop 0 ; ) | llego abajo?
  y - distfast
  30.0 <? ( pierdevida )
  drop
  drop
  ;

:+alien
'alien 'listalien p!+ >a
500.0 randmax 70.0 + a!+
-16.0 a!+
2.0 a!+ | zoom
7 4 2 vci>anim | vel cnt ini
a!+ sprites a!+ | anim sheet
2.0 randmax 1.0 -
a!+ 2.0 a!+ | vx vy
;

:horda
50 randmax 1? ( drop ; ) drop
+alien
;

|----- fondo de estrellas
#buffer * 1024 | lugar para guardar las coordenadas
#buffer> 'buffer

:+estrella | x y -
  buffer> w!+ w!+ 'buffer> ! ;

:..estrellas
$ffffff sdlColor
'buffer ( buffer> <?
  w@+ swap

```

```

        over 1 + sh >? ( 0 nip )
        over 2 - w!
        w@+ rot
        SDLPoint ) drop ;

:llenaestrellas
    256 ( 1? 1 -
        sw randmax sh randmax +estrella
    ) drop ;

|----- Juego
:juego
    0 SDLcls
    timer.
    .estrellas
    'listalien p.draw
    'listshoot p.draw
    jugador
    horda

    10 8 bat puntos "puntos:%d" sprint bprint
    10 24 bat vidas "vidas:%d" sprint bprint
    SDLredraw

    SDLkey
    >esc< =? ( exit )
    <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
    <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
    <esp> =? ( +disparo )
    drop ;

:main
    "r3sdl" 640 480 SDLinit
    bfont1
    16 16 "media/img/manual.png" ssload 'sprites !
    llenaestrellas
    200 'listalien p.ini
    200 'listshoot p.ini
    timer<
    'juego SDLshow
    SDLquit ;

: main ;

```

## Explosiones y Efectos especiales

Los efectos especiales tienen como objetivo decorar lo que está pasando en la pantalla, la mayoría de las veces no influyen en el funcionamiento del juego, por lo tanto, vamos a construirlo con una lista de sprites que solamente se crean cuando ocurre algo y se destruyen automáticamente. Vamos a agregar una lista sprites para las explosiones de los aliens, podríamos tener varios tipos de efectos especiales, y aquí si, agruparlos en una sola lista porque no van a tener interacción con el juego, son solo decorativos. Podríamos poner nubes, estrellas, efectos de chispas, quizás

necesitaríamos varias listas si utilizamos distintos planos para estos sprites, es decir, que se dibujen antes o después del jugador y de los enemigos.

En la línea 24 preguntamos si estamos en el último cuadro de la animación de la explosión para quitar el sprite.

La línea 26 agrega la explosión cuando se produce la colisión del disparo y el alien utilizando las coordenadas del mismo.

```
^r3/win/sdl2gfx.r3
^r3/lib/rand.r3
^r3/util/bfont.r3
^r3/util/arr16.r3

#sprites

#puntos 0
#vidas 3

#x 320.0 #y 380.0
#xv 0 #yv 0

#listalien 0 0 | lista de aliens
#listshoot 0 0 | lista de disparos
#listfx 0 0 | fx

:objsprite | adr -- adr
  dup >a
  a@+ int. a@+ int. | x y
  a@+ | zoom
  a@ timer+ dup a!+ | anima
  nanim a@+ sspritez
  dup 40 + @ over +! | vx
  dup 48 + @ over 8 + +! | vy
  ;

|----- fx
:explosion
  objsprite
  24 + @ nanim 10 =? ( drop 0 ; )
  drop
  ;

+:explo | y x --
  'explosion 'listfx p!+ >a
  swap a!+ a!+ | x y
  2.0 a!+ | zoom
  7 5 6 vci>anim | vel cnt ini
  a!+ sprites a!+ | anim sheet
  0 a!+ 0 a!+ | vx vy
  ;

|----- Jugador
:jugador
  x int. y int. 2.0
```

```

msec 7 >> $1 and
sprites sspritez
xv 'x +! yv 'y +!
;

:perdevida
-1 'vidas +!
vidas 0? ( exit ) drop
320.0 'x !
380.0 'y !
;

|----- Disparo
#hit
:choque? | x y i n p -- x y p
dup 8 + >a
pick4 a@+ - pick4 a@+ -
distfast 20.0 >? ( drop ; ) drop
dup 'listalien p.del
1 'puntos +!
0 'hit !
pick4 pick4 +explo
;

:bala | v --
objsprite

1 'hit !
dup @
over 8 + @ -20.0 <? ( 3drop 0 ; )
'choque? 'listalien p.mapv | 'vector list --
2drop
hit 0? ( nip ; ) drop
drop
;

:+disparo
'bala 'listshoot p!+ >a
x a!+ y 16.0 - a!+ | x y
1.0 a!+ | zoom
8 2 10 vci>anim | vel cnt ini
a!+ sprites a!+ | anim sheet
0 a!+ -3.0 a!+ | vx vy
;

|----- Alien
:alien | v --
objsprite

dup @ x -
over 8 + @
500.0 >? ( 3drop 0 ; ) | llevo abajo?
y - distfast
30.0 <? ( perdevida )
drop

```

```

        drop
        ;

:+alien
    'alien 'listalien p!+ >a
    500.0 randmax 70.0 + a!+
    -16.0 a!+
    2.0 a!+      | zoom
    7 4 2 vci>anim | vel cnt ini
    a!+  sprites a!+      | anim sheet
    2.0 randmax 1.0 -
    a!+ 2.0 a!+ | vx vy
    ;

:horda
    50 randmax 1? ( drop ; ) drop
    +alien
    ;

|----- fondo de estrellas
#buffer * 1024      | lugar para guardar las coordenadas
#buffer> 'buffer

:+estrella | x y -
    buffer> w!+ w!+ 'buffer> ! ;

:.estrellas
    $ffffff sdlColor
    'buffer ( buffer> <?
        w@+ swap
        over 1 + sh >? ( 0 nip )
        over 2 - w!
        w@+ rot
        SDLPoint ) drop ;

:llenaestrellas
    256 ( 1? 1 -
        sw randmax sh randmax +estrella
    ) drop ;

|----- Juego
:juego
    0 SDLcls
    timer.
    .estrellas
    'listalien p.draw
    'listshoot p.draw
    'listfx p.draw
    jugador
    horda

    10 8 bat puntos "puntos:%d" sprint bprint
    10 24 bat vidas "vidas:%d" sprint bprint
    SDLredraw

    SDLkey

```




```

>esc< =? ( exit )
<le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
<ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
<esp> =? ( +disparo )
drop ;

:main
  "r3sdl" 640 480 SDLinit
  bfont1
  16 16 "media/img/manual.png" ssload 'sprites !
  llenaestrellas
  200 'listalien p.ini
  200 'listshoot p.ini
  200 'listfx p.ini
  timer<
  'juego SDLshow
  SDLquit ;

: main ;

```



## Sonidos, Música

El sonido y la música agregan otro nivel de realismo al juego, para esto vamos a utilizar la biblioteca de música y sonidos de SDL2

^r3/win/sdl2mixer.r3

Al inicio del programa vamos a cargar los archivos de música o sonidos que se van a disparar durante la ejecución.

Para esto vamos a definir variables para guardar las referencias a estos elementos, podemos cargar un sonido o un tema musical completo

Las palabras para cargar los archivos de sonido son:

```
::Mix_loadMus | filename -- referencia
```

Carga en memoria un archivo de sonido que será usado como música, generalmente de larga duración.

```
::Mix_LoadWAV | filename -- referencia
```

Carga un archivo de sonido que será usado como efecto, generalmente de corta duración. Los principales formatos de sonidos son .wav, .mp3 o .ogg, tanto para música como para sonido.

```
::Mix_PlayMusic | musica loop -
```

Reproduce un archivo de sonido, generalmente largo, música es la referencia que se cargó con Mix\_LoadMus y loop será la cantidad de veces que se va a reproducir, con -1 se repetirá infinitamente, hasta que se apague, terminando la reproducción o comenzando otra música.

```
::SNDplay | sonido -
```

Reproduce un sonido, lo que necesita de la pila es la referencia que se cargó con Mix\_LoadWAV, esta biblioteca mantiene varios canales de sonido, quiere decir que mientras suene la música pueden aparecer varios efectos de sonido, como explosiones o disparos y sonarán todos a la vez. La biblioteca posee más posibilidades pero con solamente estas palabras es posible hacer lo básico.

```
^r3/win/sdl2gfx.r3
^r3/win/sdl2mixer.r3
^r3/lib/rand.r3
^r3/util/bfont.r3
^r3/util/arr16.r3

#sprites          | hoja de sprites
#snd_shoot         | sonido de disparo
#snd_explode       | sonido de explosion

#puntos 0
#vidas 3

#x 320.0 #y 380.0
#xv 0 #yv 0

#listalien 0 0 | lista de aliens
#listshoot 0 0 | lista de disparos
#listfx 0 0 | fx

:objsprite | adr -- adr
  dup >a
  a@+ int. a@+ int. | x y
  a@+                | zoom
  a@ timer+ dup a!+ | anima
  nanim a@+ sspritez
  dup 40 + @ over +! | vx
```

```

        dup 48 + @ over 8 + +! | vy
        ;

|----- fx
:explosion
    objsprite
    24 + @ nanim 10 =? ( drop 0 ; )
    drop
    ;

:+explo      | y x --
    'explosion 'listfx p!+ >a
    swap a!+ a!+| x y
    2.0 a!+      | zoom
    7 5 6 vci>anim | vel cnt ini
    a!+  sprites a!+      | anim sheet
    0 a!+ 0 a!+ | vx vy
    snd_explode SNDplay
    ;

|----- Jugador
:jugador
    x int. y int. 2.0
    msec 7 >> $1 and
    sprites sspritez
    xv 'x +! yv 'y +!
    ;

:pierdevida
    -1 'vidas +!
    vidas 0? ( exit ) drop
    320.0 'x ! 380.0 'y !
    ;

|----- Disparo
#hit
:choque? | x y i n p -- x y p
    dup 8 + >a
    pick4 a@+ - pick4 a@+ -
    distfast 20.0 >? ( drop ; ) drop
    dup 'listalien p.del
    1 'puntos +!
    0 'hit !
    pick4 pick4 +explo
    ;

:bala | v --
    objsprite

    1 'hit !
    dup @
    over 8 + @ -20.0 <? ( 3drop 0 ; )
    'choque? 'listalien p.mapv | 'vector list --
    2drop
    hit 0? ( nip ; ) drop

```

```

        drop
        ;

:+disparo
    'bala 'listshoot p!+ >a
    x a!+ y 16.0 - a!+ | x y
    1.0 a!+      | zoom
    8 2 10 vci>anim | vel cnt ini
    a!+ sprites a!+ | anim sheet
    0 a!+ -3.0 a!+ | vx vy
    snd_shoot SNDplay
    ;

|----- Alien
:alien | v --
    objsprite

    dup @ x -
    over 8 + @
    500.0 >? ( 3drop 0 ; ) | llego abajo?
    y - distfast
    30.0 <? ( pierdevida )
    drop
    drop
    ;

:+alien
    'alien 'listalien p!+ >a
    500.0 randmax 70.0 + a!+
    -16.0 a!+
    2.0 a!+      | zoom
    7 4 2 vci>anim | vel cnt ini
    a!+ sprites a!+ | anim sheet
    2.0 randmax 1.0 -
    a!+ 2.0 a!+ | vx vy
    ;

:horda
    50 randmax 1? ( drop ; ) drop
    +alien
    ;

|----- fondo de estrellas
#buffer * 1024          | lugar para guardar las coordenadas
#buffer> 'buffer

:+estrella | x y -
    buffer> w!+ w!+ 'buffer> ! ;

:.estrellas
    $ffffff sdlColor
    'buffer ( buffer> <?
        w@+ swap
        over 1 + sh >? ( 0 nip )
        over 2 - w!
        w@+ rot

```

```

        SDLPoint ) drop ;

:llenaestrellas
  256 ( 1? 1 -
    sw randmax sh randmax +estrella
  ) drop ;

|----- Juego
:juego
  0 SDLcls
  timer.
  .estrellas
  'listalien p.draw
  'listshoot p.draw
  'listfx p.draw
  jugador
  horda

  10 8 bat puntos "puntos:%d" sprint bprint
  10 24 bat vidas "vidas:%d" sprint bprint
  SDLredraw

  SDLkey
  >esc< =? ( exit )
  <le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
  <ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
  <esp> =? ( +disparo )
  drop ;

:jugar
  'listalien p.clear
  'listshoot p.clear
  'listfx p.clear
  0 'puntos ! 3 'vidas !
  320.0 'x ! 380.0 'y !
  'juego SDLShow ;

:main
  "r3sdl" 640 480 SDLinit
  bfont1
  16 16 "media/img/manual.png" ssload 'sprites !
  "media/snd/shoot.mp3" Mix_LoadWAV 'snd_shoot !
  "media/snd/explode.mp3" Mix_LoadWAV 'snd_explode !
  "media/ttf/Roboto-bold.ttf" 40 TTF_OpenFont immSDL
  llenaestrellas
  200 'listalien p.ini
  200 'listshoot p.ini
  200 'listfx p.ini
  timer<
  jugar
  SDLquit ;

: main ;

```

## Menu de inicio

Utilizando la libreria util/sdlgui.r3 podemos dibujar botones en una pantalla de inicio.

```
^r3/win/sdl2gfx.r3
^r3/win/sdl2mixer.r3
^r3/lib/rand.r3
^r3/util/bfont.r3
^r3/util/arr16.r3
^r3/util/sdlgui.r3

#sprites          | hoja de sprites
#snd_shoot         | sonido de disparo
#snd_explode| sonido de explosion

#puntos 0
#vidas 3

#x 320.0 #y 380.0
#xv 0 #yv 0

#listalien 0 0 | lista de aliens
#listshoot 0 0 | lista de disparos
#listfx 0 0 | fx

:objsprite | adr -- adr
  dup >a
  a@+ int. a@+ int. | x y
  a@+                | zoom
  a@ timer+ dup a!+ | anima
  nanim a@+ sspritez
  dup 40 + @ over +! | vx
  dup 48 + @ over 8 + +! | vy
  ;

|----- fx
:explosion
  objsprite
  24 + @ nanim 10 =? ( drop 0 ; )
  drop
  ;

:+explo      | y x --
  'explosion 'listfx p!+ >a
  swap a!+ a!+ | x y
  2.0 a!+                | zoom
  7 5 6 vci>anim | vel cnt ini
  a!+  sprites a!+                | anim sheet
  0 a!+ 0 a!+ | vx vy
  snd_explode SNDplay
  ;
```

```

|----- Jugador
:jugador
  x int. y int. 2.0
  msec 7 >> $1 and
  sprites sspritez
  xv 'x +! yv 'y +!
  ;

:pierdevida
  -1 'vidas +!
  vidas 0? ( exit ) drop
  320.0 'x ! 380.0 'y !
  ;

|----- Disparo
#hit
:choque? | x y i n p -- x y p
  dup 8 + >a
  pick4 a@+ - pick4 a@+ -
  distfast 20.0 >? ( drop ; ) drop
  dup 'listalien p.del
  1 'puntos +!
  0 'hit !
  pick4 pick4 +explo
  ;

:bala | v --
  objsprite

  1 'hit !
  dup @
  over 8 + @ -20.0 <? ( 3drop 0 ; )
  'choque? 'listalien p.mapv | 'vector list --
  2drop
  hit 0? ( nip ; ) drop
  drop
  ;

:+disparo
  'bala 'listshoot p!+ >a
  x a!+ y 16.0 - a!+ | x y
  1.0 a!+ | zoom
  8 2 10 vci>anim | vel cnt ini
  a!+ sprites a!+ | anim sheet
  0 a!+ -3.0 a!+ | vx vy
  snd_shoot SNDplay
  ;

|----- Alien
:alien | v --
  objsprite

  dup @ x -
  over 8 + @

```

```

500.0 >? ( 3drop 0 ; ) | llego abajo?
y - distfast
30.0 <? ( pierdevida )
drop
drop
;

:+alien
'alien 'listalien p!+ >a
500.0 randmax 70.0 + a!+
-16.0 a!+
2.0 a!+ | zoom
7 4 2 vci>anim | vel cnt ini
a!+ sprites a!+ | anim sheet
2.0 randmax 1.0 -
a!+ 2.0 a!+ | vx vy
;

:horda
50 randmax 1? ( drop ; ) drop
+alien
;

|----- fondo de estrellas
#buffer * 1024 | lugar para guardar las coordenadas
#buffer> 'buffer

:+estrella | x y -
buffer> w!+ w!+ 'buffer> ! ;

:.estrellas
$ffffff sdlColor
'buffer ( buffer> <?
w@+ swap
over 1 + sh >? ( 0 nip )
over 2 - w!
w@+ rot
SDLPoint ) drop ;

:llenaestrellas
256 ( 1? 1 -
sw randmax sh randmax +estrella
) drop ;

|----- Juego
:juego
0 SDLcls
timer.
.estrellas
'listalien p.draw
'listshoot p.draw
'listfx p.draw
jugador
horda

10 8 bat puntos "puntos:%d" sprint bprint

```



```

10 24 bat vidas "vidas:%d" sprint bprint
SDLredraw

SDLkey
>esc< =? ( exit )
<le> =? ( -1.0 'xv ! ) >le< =? ( 0 'xv ! )
<ri> =? ( 1.0 'xv ! ) >ri< =? ( 0 'xv ! )
<esp> =? ( +disparo )
drop ;

:jugar
'listalien p.clear
'listshoot p.clear
'listfx p.clear
0 'puntos ! 3 'vidas !
320.0 'x ! 380.0 'y !
'juego SDLShow ;

:menu
0 SDLcls
timer.
.estrellas
immgui

0 100 immat
640 immwidth
$ffffff 'immcolortex !
"Xenofobia Extraterrestre" immLabelC
320 immwidth
160 300 immat

$7f00 'immcolorbtn !
'jugar "Jugar" immbtn
immdn
$7f0000 'immcolorbtn !
'exit "Salir" immbtn

SDLredraw
SDLkey
>esc< =? ( exit )
drop ;

:main
"r3sdl" 640 480 SDLinit
bfont1
16 16 "media/img/manual.png" ssload 'sprites !
"media/snd/shoot.mp3" Mix_LoadWAV 'snd_shoot !
"media/snd/explode.mp3" Mix_LoadWAV 'snd_explode !
"media/ttf/Roboto-bold.ttf" 40 TTF_OpenFont immSDL
llen aestrellas
200 'listalien p.ini
200 'listshoot p.ini
200 'listfx p.ini
timer<
'menu SDLshow

```

	<pre>SDLquit ; : main ;</pre>
--	-----------------------------------